
Stata+α 

「新型コロナウィルスの感染データを Stata で入手する 

(国内編)」 

前回、「新型コロナウィルスの感染データを Stata で入手する」ではジョンズ・ホプキンス

大学が公開しているは各国のコロナウィルス感染者数を Stata でアクセスし、Stata 形式に

変換して整理しました。今回は、「COVID-19 Japan 新型コロナウイルス対策ダッシュボー

ド」で公開されている都道府県別の感染情報にアクセスし、Stata 形式に変換します。 

1. データのインポート

「 COVID-19 Japan 新 型 コ ロ ナ ウ イ ル ス 対 策 ダ ッ シ ュ ボ ー ド 」

(https://www.stopcovid19.jp/)は、前回紹介したジョンズ・ホプキンス大学と同じように

https://www.stopcovid19.jp/data/covid19japan/2020-03-22.csv という形で csv ファイルを

公開しています。各ファイルは、YYYY-MM-DD.csv となっています。1 日分の csv ファイ

ルを import delimited コマンドで Stata に取り込んでみましょう。都道府県名は漢字で

入力されているため、encoding()オプションで、文字コードを指定します。 

1 import delimited "https://www.stopcovid19.jp/data/covid19japan/2020-

03-22.csv", encoding(UTF-8)

データエディタには次のようなデータセットがインポートされます。 

https://www.lightstone.co.jp/stata/ug000.html#import
https://www.stopcovid19.jp/
https://www.stopcovid19.jp/data/covid19japan/2020-03-22.csv

2. ファイルの一括ダウンロード

公開されている csv ファイルを一括でダウンロードするには、import delimitedを何度も

繰り返します。この操作は、前回と同様です。繰り返しには同じく forvaluesと localコ

マンドを利用します。次のコマンドを do ファイルエディタに貼り付けて実行しましょう。 

1 

2 

3 

local URL = "https://www.stopcovid19.jp/data/covid19japan/" 

forvalues month = 1/12 { 

forvalues day = 1/31 { 

local month = string(`month', "%02.0f") 

local day = string(`day', "%02.0f") 

local year = "2020" 

local today = "`year'-`month'-`day'" 

local FileName = "`URL'`today'.csv" 

clear 

capture import delimited "`FileName'", encoding(UTF-

8)

if _rc == 0 { 

generate date = "`today'" 

capture confirm variable ndeath 

if _rc == 0 { 

    destring ndeath, replace force 

} 

capture confirm variable 

nheavycurrentpatients 

if _rc == 0 { 

    destring nheavycurrentpatients, replace 

force 

} 

} 

capture save "`today'", replace 

   } 

} 

基本的には、前回の「3. 複数ファイルを一括ダウンロード」と同様ですが、大きく異なっ

ている点は、11-21 行目まで一連のコマンドです。11 行目以降は ifコマンドの括弧で囲ま

れています。10 行目の capture import delimitedコマンドは、ファイルのインポート

が成功すれば、システム変数_rcに 0 を代入し、エラーが発生すると該当のエラーコードを

代入します。つまり csv ファイルの読み込みが成功した場合のみ、12-21 行目までが実行さ

れます。 

まず、ダウンロードしたデータセットにはファイル名にしか調査した日付がありませんの

で、12 行目で、generateコマンドを使用し日付を新しい変数として作成します。 

13-19 行では、5 月 8 日付けデータセットから新たに集計が開始された 2 変数、ndeath と

nheavycurrentpatients が文字列変数として読み込まれてしまう問題に対処しています。

13行目の capture confirm ndeathコマンドで読み込んだデータセットに変数 ndeathが

あれば、15 行目の destring コマンドで数値変数に変換します。17-19 行も同様の処理を

変数 nheavycurrentpatientsに対して行います。

このようにして、変数を整えたら、22行目の capture saveコマンドで YYYY-MM-DD.dta

というファイル名の Stata 形式ファイルとして保存します。

3. ファイルの統合

ダウンロードしたファイルを appendコマンドで１つの dta ファイルに統合します。append

も forvaluesコマンドと組み合わせて繰り返します。こちらも前回の「4. データをまとめ

る」で使用したものをほぼ同様です。8 行目の年月日の並び方が異なっていることにご注意

ください。 

1 

2 

3 

4 

5 

6 

7 

8 

9 

10 

clear 

forvalues month = 1/12 { 

   forvalues day = 1/31 { 

local month = string(`month', "%02.0f") 

local day = string(`day', "%02.0f") 

local year = "2020" 

local today = "`year'-`month'-`day'" 

capture append using "`today'" 

   } 

} 

統合したデータセットは次のようになります。 

4. データを整える

3.で統合したデータセットを確認すると、都道府県名の変数 nameに入力ミスが見つかりま

す。これは replaceコマンドで修正します。

1 replace name = "Ehime" if name == "Eihime" 

このデータセットは、時間変数 dateと ID 変数 nameまたは name_jpを持つパネル形式で

す。変数 date と name を文字列から数値に変換して、パネルデータとして宣言しましょう。

変数 name または name_jp のような文字列変数に数値を割り振って数値(カテゴリ)変数を

作成するには、encode コマンドを使用します。ここでは、変数 name から新しい変数

prefectureを作成しています。 

1 encode name, generate(prefecture) 

続いて、文字列となっている変数 dateから generateコマンドで数値変数 date2を新規作

成します。数値変数 date2 はそのままでは人間が認識できる形式ではないので、2 行目の

formatコマンドで表示形式を変更します。 

1 

2 

generate date2 = daily(date, "YMD") 

format %tdCCYY-NN-DD date2 

表示形式の変更は、変数マネージャから行うこともできます。変数マネージャを開き、date2

を選択して「フォーマット」の「作成」ボタンをクリックします。「書式作成」ダイアログ

でフォーマットを変更したら、OK をクリックして閉じ、変数マネージャの「適用」をクリ

ックします。 

変数マネージャの画面は次のようになります。 

時間変数と ID 変数の準備ができたら、xtset コマンドでパネルデータであることを Stata

に認識させます。 

1 xtset prefecture date2 

結果ウィンドウには次のように、データセットの構造が表示されます。パネル ID は

prefecture、期間は 3 月 18 日から 8 月 1 日の 1 日毎のデータです。 

変数 npatientsは累積感染者数です。1 日の新規感染者数を求めるには前日の npatients

との差を計算します。Stata では差分を d.演算子で簡単に求めることできます。d.演算子と

generateコマンドで 1 日あたりの新規感染者数 newcaseを作成します。 

1 generate newcase = d.npatients 

新規感染者数は曜日によって変動があることがわかっています。これをグラフで確認して

みましょう。東京都の新規感染者数を折れ線グラフにします、if オプションで対象を制限

しています。 

1 twoway line newcase date2 if name == "Tokyo", sort 

 delta:  1 day

 time variable:  date2, 2020-03-18 to 2020-08-01

 panel variable:  prefecture (strongly balanced)

. xtset prefecture date2

グラフから周期的な変動があること確認できました。このような変動を抑えて、数値の変化

を調べるには、移動平均を求めます。変数 newcaseの直近 1 週間の平均を新しい変数とし

て作成しましょう。Stata では前期の値を参照するために、l.演算子を使用します。使用し

ているのデータセットは日次データなので、l.newcaseは前日、l2.newcaseは 2日前の新

規感染者数を参照しています。パネル ID ごとに計算するために、bysort プリフィックス

コマンドを使用しています。

1 bysort prefecture (date2): generate ma = (l6.newcase + l5.newcase + 

l4.newcase + l3.newcase + l2.newcase + l.newcase + newcase) / 7 

新しく作成した直近 1 週間の感染者数平均を示す変数 maを、先の折れ線グラフに追加しま

す。 

1 twoway (line newcase date2 if name == "Tokyo", sort) (line ma date2 

if name == "Tokyo", sort) 

移動平均をプロットしたグラフは次のようになります。周期的な変動がスムージングされ

ていることがわかります。 

5. まとめ

今回は、「COVID-19 Japan 新型コロナウイルス対策ダッシュボード」が公開している日本

国内の新型コロナウィルス感染情報を、Stata からアクセスして保存する方法を紹介しまし

た。ウェブ上で公開されている csv ファイルは、import delimited “URL”で Stata に取り

込みます。時系列データやパネルデータでは、差分を取る場合は d.演算子を、前期の値を

参照するには l.演算子を使用します。このほかに後期の値を参照する f.演算子も用意され

ています。データセット構造に関係なく 1 つ上のセルを参照する場合は、変数名[_n-1]と

入力します。 

